

Guía básica de LINGO

software de optimización matemática

www.lindo.com

I. Entorno de Desarrollo de modelos en el software LINGO

¿Qué es LINGO?

LINGO (**L**inear, **I**nteractive, and **G**eneral **O**ptimizer), es un lenguaje de modelación matemática, que provee un entorno en el cual se puede desarrollar, correr y modificar modelos matemáticos.

LINGO es una herramienta fácil de utilizar para desarrollar grandes modelos de optimización lineal y no lineal.

Uso de Conjuntos (SETS)

Comúnmente los grandes modelos requieren expresar grupos de muchas restricciones, las cuales mantienen una estructura similar. Dada esta característica, LINGO, permite manejarlo como conjuntos (SET) de información y realizar las operaciones de una manera más eficiente. Usando el estilo del modelo directo, usted necesitará encargarse de tipear de manera repetitiva cada término de cada restricción. Los SETS le permiten definir y trabajar con grupos de objetos que necesitan ser procesados de formas similares. En efecto, los mismos objetos pueden incluso ser otros conjuntos. Los SETS son el fundamento del Lenguaje de Modelación de LINGO. Con una definición de conjuntos puede escribir una serie de restricciones similares en una sentencia simple y expresar formulaciones largas y complejas consistentemente. Esto le permite expresar sus modelos más grandes muy rápidamente, y de cierta manera es fácil de leer y entender.

LA SECCION SET DE UN MODELO


Consideraciones sobre el uso de conjuntos (SETS)

- Es la característica más poderosa en LINGO para modelar sistemas grandes.
- Permite expresar muchas restricciones similares en pocas líneas.
- Antes de usar conjuntos en un modelo LINGO, tenemos que definirlos en la sección de conjuntos del modelo.
- La sección de conjuntos comienza con la palabra clave “ **SETS:**” (incluyendo los dos puntos) y termina con la palabra clave “**ENDSETS**”.
- Un modelo puede, no tener la sección de conjuntos, una sola o múltiples secciones.
- Una sección de conjuntos puede aparecer en cualquier lugar del modelo.
- La única restricción es que se debe definir cualquier conjunto y sus atributos antes de que los mismos sean referenciados en las restricciones del modelo.
- El nombre del conjunto es cualquiera que el usuario seleccione, comenzando con un carácter alfabético, seguido de hasta 31 caracteres alfanuméricos o el underline “_”.

- La lista de miembros del conjunto es aquella que constituyen el conjunto, estos pueden ser listados en forma explícita o implícita.
- Los miembros del conjunto pueden tener cero o mas atributos especificados en la lista de atributos de la definición del conjunto.
- Un atributo es simplemente alguna propiedad que cada uno de los miembros del conjunto posee.

Los SETS (conjuntos) son simplemente grupos de objetos seleccionados. Un SET puede ser una lista de productos, tareas, o stocks. Cada elemento del conjunto puede tener una o más características asociadas con éste. Llamaremos a estas características atributos. Por ejemplo, un SET de producto puede tener un atributo que lista el precio de cada producto.

Tipos de Conjuntos



Conjunto Primitivo:

Los conjuntos primitivos son los objetos fundamentales en un modelo y no pueden ser desmenuzados en componentes más pequeños.

La definición de un conjunto primitivo requiere:

- El nombre del conjunto.
- Sus miembros (objetos contenidos en el conjunto).
- Opcionalmente, cualquier atributo que los miembros del conjunto puedan tener.

Sintaxis:

nombre_conjunto / lista_miembros / [: lista_atributos]

SUCURSAL / AREQUIPA, CUZCO, TRUJILLO / : INGRESOS;

En algunos casos de definición de conjuntos no es necesario definir atributos asociados a los miembros del conjunto, por ejemplo el caso anterior podría quedar así:

SUCURSAL / AREQUIPA, CUZCO, TRUJILLO / ; ;

Los miembros de los dos ejemplos anteriores han sido definidos por extensión.

nombre_conjunto / 1 ..n / [: lista_atributos]

PROYECTO / 1 .. 3 / : RENTABILIDAD;

Los miembros del ejemplo anterior han sido definidos por compresión. Ambos casos son validos.

Conjunto Derivado Denso:

Los conjuntos derivados, son creados a partir de otros conjuntos (uno o más de otros conjuntos). Estos conjuntos componentes son referenciados como los padres del conjunto derivado y pueden ser primitivos o derivados.

Para definir un conjunto derivado se requiere:

- El nombre del conjunto
- Sus conjuntos padres
- Opcionalmente, sus miembros y
- Opcionalmente, cualquier atributo que los miembros del conjunto puedan tener.

Se tiene la siguiente sintaxis:

nombre_conjunto (Lista_conjuntos_padres)
[/especificaciones_lista_miembros/] [:lista_atributos];

Consideremos la siguiente sección a manera de ejemplo.

SETS:

```
SUCURSAL / AREQUIPA, CUZCO /;;  
PROYECTO / PROY1, PROY2/;;  
PERIODO / 1..2 /;;  
SUC_PROY_PER(SUCURSAL, POYECTO, PERIODO): TASA, INVMAXIMA;  
ENDSETS
```

- El “nombre_conjunto” es un nombre cualesquiera que se define para denominar al conjunto, en lo posible algo que sugiera de que conjuntos se deriva, o el uso que se le va a dar al conjunto.
- La “Lista_conjuntos_padres” es una lista de conjuntos previamente definidos, separados por comas.
- Sin especificar los elementos de “especificaciones_lista_miembros”, LINGO construye todas las combinaciones de miembros de cada uno de los conjuntos padres para crear los miembros del nuevo conjunto derivado.

Miembros del conjunto SUC_PROY_PER

<u>Indice</u>	<u>Miembro</u>
1	(AREQUIPA, PROY1, 1)
2	(AREQUIPA, PROY1, 2)
3	(AREQUIPA, PROY2, 1)
4	(AREQUIPA, PROY2, 2)
5	(CUZCO, PROY1, 1)
6	(CUZCO, PROY1, 2)
7	(CUZCO, PROY2, 1)
8	(CUZCO, PROY2, 2)

Los conjuntos SUCURSAL, PROYECTO y PERIODO son conjuntos primitivos que conforman el conjunto derivado SUC_PROY_PER. Tomando todas las combinaciones de los miembros de cada uno de los tres conjuntos padres, LINGO construye el conjunto SUC_PROY_PER:

- La lista de miembros es opcional y es usada cuando se desea limitar el conjunto a algunos subconjuntos del conjunto completo.
- Una lista de miembros de un conjunto derivado puede ser construido usando: Una lista de miembros explícita o Como un filtro de miembros.

También se podría generar un SET derivado a partir de un solo tipo de SET padre:

SETS:

```
SUCURSAL / AREQUIPA, CUZCO, TRUJILLO /;
CONEXION(SUCURSAL, SUCURSAL): DISTANCIA;
ENDSETS
```

Miembros del conjunto CONEXION

<u>Indice</u>	<u>Miembro</u>
1	(AREQUIPA, AREQUIPA)
2	(AREQUIPA, CUZCO)
3	(AREQUIPA, TRUJILLO)
4	(CUZCO, AREQUIPA)
5	(CUZCO, CUZCO)
6	(CUZCO, TRUJILLO)
7	(TRUJILLO, AREQUIPA)
8	(TRUJILLO, CUZCO)
9	(TRUJILLO, TRUJILLO)

Conjunto Derivado No Denso (Usando una Lista de Miembros Explícita)

nombre_conjunto (Lista_conjuntos_padres)

[/especificaciones_lista_miembros/] [:lista_atributos];

```
SUC_PROY_PER (SUCURSAL, POYECTO, PERIODO)
/ AREQUIPA PROY1 1, AREQUIPA PROY2 2, CUZCO PROY2 1 /;
TASA, INVMAXIMA;
```

SUC_PROY_PER tendrá solo tres miembros del conjunto disperso.
(AREQUIPA, PROY1, 1), (AREQUIPA, PROY2, 2) y (CUZCO, PROY2, 1)

Conjunto Derivado No Denso (Usando un Filtro de Miembros)

- Nos permite definir una lista de miembros derivada del conjunto.
- Involucra la especificación de una condición lógica que cada miembro potencial del conjunto debe satisfacer para su inclusión en el conjunto.
- Suponga que tiene un conjunto llamado CAMIONES y cada camión tiene un atributo llamado CAPACIDAD. Además, desea generar un subconjunto solamente de aquellos camiones que son capaces de transportar grandes cargas.
- Ud. podría usar una lista de miembros explícita, e ingresar cada uno de los camiones que que pueden transportar grandes cargas, pero, por que hacer todo ese trabajo cuando se puede usar un filtro de miembros.

```
CAMIONES_PESADOS(CAMIONES) | CAPACIDAD(&1) #GT# 50000;
```

- El conjunto CAMIONES_PESADOS ha sido derivado del conjunto padre CAMIONES. La barra vertical “|” es usada para marcar el inicio de un filtro de miembros (Ver ejemplo en LINGO).
- Los operadores lógicos reconocidos por LINGO incluyen:
 - #EQ# Igual
 - #NE# No igual
 - #GE# Mayor que o igual a
 - #LT# Menor que
 - #LE# Menor que o igual a
- El filtro de miembros permite solo aquellos camiones que tienen una capacidad de carga (CAPACIDAD(&1)) mayor que (#GT) 50,000 en el conjunto CAMIONES_PESADOS.
- El simbolo &1 en el filtro es conocido como un ubicador de indice del conjunto.
- El primer conjunto primitivo padre es colocado en &1, el segundo en &2 y asi. En este ejm. Solo tenemos un un conjunto padre, de modo que &2 no tiene sentido.

II DESARROLLANDO MODELOS

El siguiente modelo, ilustra el uso del SET derivado, las sentencias @SUM y @FOR con filtros.

PROBLEMA : MODELO DE ASIGNACION DE PROYECTO – (MODELO A)

Una compañía dispone de 35 millones para distribuirlos el próximo año entre sus sucursales (Arequipa, Cuzco, Trujillo). Debido a compromisos de la estabilidad, el nivel de empleados y por otras razones la compañía ha establecido un nivel mínimo de asignación de fondos para cada sucursal. Estos fondos mínimos son de 8, 10 y 15 millones de dólares respectivamente. Debido a la naturaleza de su operación, la sucursal Cuzco no puede utilizar más de 20 millones de dólares. Cada sucursal tiene la oportunidad de dirigir distintos proyectos con los fondos que recibe. Para cada proyecto se ha establecido una tasa de ganancia (como un porcentaje de la inversión). Por otra parte, algunos de los proyectos permiten solo una inversión limitada. A continuación se dan los datos para cada proyecto.

Sucursal	Proyecto	Tasa de Utilidad (%)	Inversión Máxima
Arequipa	1	8	10
	2	6	5
	3	7	9
Cuzco	1	5	7
	2	8	10
	3	9	4
Trujillo	1	10	6
	2	6	12
	3	15	6

Solución:

Primero veremos el modelo antes de investigar como trabaja.

Formulación Algebraica del modelo

Definición de variables:

$X_{(i,j)}$ = Cantidad asignada a la sucursal i para el proyecto j .

Definición de parámetros:

Mínimo(i) = Nivel mínimo de asignación de fondos para la sucursal i .

Máximo(i) = Nivel máximo de asignación de fondos para la sucursal i .

InvMaxima(i, j) = Inversión máxima en la sucursal i para el proyecto j .

PresupuestoTotal = Presupuesto total disponible.

Tasa(i, j) = Tasa de rentabilidad del proyecto j en la sucursal i .

Restricción de uso mínimo de presupuesto por sucursal.

$$\sum_{j=1}^3 X_{(i,j)} \geq \text{Mínimo}(i) \quad \forall i = 1,2,3$$

Restricción de uso máximo de presupuesto por sucursal.

$$\sum_{j=1}^3 X_{(i,j)} \leq \text{Máximo}(i) \quad \forall i = 2 \quad (\text{Solo para Cuzco})$$

Restricción de inversión máxima por proyecto en cada sucursal.

$$X_{(i,j)} \leq \text{InvMaxima}(i,j) \quad \forall i = 1,2,3; \quad \forall j = 1,2,3$$

Restricción de presupuesto disponible.

$$\sum_{i=1}^3 \sum_{j=1}^3 X_{(i,j)} \leq \text{PresupuestoTotal}$$

Función Objetivo

$$\text{MAX} = \sum_{i=1}^3 \sum_{j=1}^3 \text{Tasa}(i, j) * X_{(i,j)}$$

Pasos para identificar y relacionar los elementos que participan en la estructura de un modelo en LINGO

Para los principiantes en el uso de LINGO les recomendamos hacer este análisis para poder entender las relaciones entre los elementos y saber como definir la estructura de un modelo en un entorno LINGO. Si usted ya es un usuario de LINGO podría obviar estos pasos.

1. Describir los elementos que participan en el modelo.

(i) = Índice de sucursal.

(j) = Índice de proyectos.

Mínimo(i) = Nivel mínimo de asignación de fondos para la sucursal i .

Máximo(i) = Nivel máximo de asignación de fondos para la sucursal i .

InvMaxima(i, j) = Inversión máxima en la sucursal i para el proyecto j .

PresupuestoTotal = Presupuesto total disponible.

TasaRentabilidad(i, j) = Tasa de rentabilidad del proyecto i en la sucursal j .

$X_{(i,j)}$ = Cantidad asignada a la sucursal i para el proyecto j .

2. Matriz de elementos que participan en el modelo y sus relaciones, Indicar cuales son parámetros y variables.

Clasific.	ELEMENTOS	Sucursal (i)	Proyecto (j)	Tasa	Inv. Máximo	Mínimo	Máximo	Presup	Asignación
Índice	Sucursal (i)								
Índice	Proyecto (j)								
Parámetro	Tasa	X	X						
Parámetro	Inv. Máxima	X	X						
Parámetro	Mínimo	X							
Parámetro	Máximo	X							
Parámetro	Presupuesto								
Variable	$X_{(i,j)}$	X	X						

3. Relacionar los elementos (“depende de”).

De la matriz anterior podemos observar que:

- ◆ Sucursal(i) no dependen de ningún elemento superior, por lo tanto es un elemento básico.
- ◆ Proyecto(j) no dependen de ningún elemento superior, por lo tanto es un elemento básico.
- ◆ Tasa depende de los elementos Sucursal(i) y Proyecto(j).
- ◆ Inv. Máxima depende de los elementos Sucursal(i) y Proyecto(j).
- ◆ Mínimo depende solo del elemento Sucursal(i).
- ◆ Máximo depende solo del elemento Sucursal(i).
- ◆ Presupuesto es un parámetro aislado y no depende de ningún elemento.

- ◆ $X_{(i,j)}$ es una variable que depende de los elementos Sucursal(i) y Proyecto(j).

4. Relacionar los elementos (“depende de”) y plantear una estructura de SET.

Sucursal(i), no depende de ningún elemento superior, por lo tanto es un elemento básico, que se define como un conjunto primitivo, para nuestro ejemplo lo llamaremos el set SUCURSAL definido por extensión (Arequipa, Cuzco, Trujillo).

```
SETS:
SUCURSAL /AREQUIPA,CUZCO,TRUJILLO/ ;
ENDSETS
```

Proyecto(j), no depende de ningún elemento superior, por lo tanto es un elemento básico, que se define como un conjunto primitivo, para nuestro ejemplo lo llamaremos el set PROYECTO, compuesto por tres proyectos, y lo definiremos por comprensión.

```
SETS:
SUCURSAL /AREQUIPA,CUZCO,TRUJILLO/ ;
PROYECTO /1...3/ ;
ENDSETS
```

Los elementos Tasa, Inv. Máxima y $X_{(i,j)}$, depende de los elementos Sucursal(i) y Proyecto(j) (definidos anteriormente como conjuntos), por lo tanto es necesario crear una estructura de un set derivado que permita enlazar ambos conjuntos; para nuestro ejemplo lo llamaremos el set SUCURSAL_PROYECTO, compuesto por los conjuntos SUCURSAL y PROYECTO, y lo definiremos como un conjunto derivado denso.

Así mismo los elementos Tasa, Inv. Máxima y $X_{(i,j)}$, serán llamados atributos del conjunto derivado SUCURSAL_PROYECTO, y los definiremos como TASA, INVMAXIMA y ASIGNACION.

```
SETS:
SUCURSAL /AREQUIPA,CUZCO,TRUJILLO/ ;
PROYECTO /1...3/ ;
SUCURSAL_PROYECTO(SUCURSAL, PROYECTO): TASA, INVMAXIMA, ASIGNACION ;
ENDSETS
```

Los elementos mínimo y máximo depende solo del elemento Sucursal(i), por lo tanto serán llamados atributos del conjunto primitivo SUCURSAL, y los definiremos como MINIMO y MAXIMO. Finalmente la sección SETS quedara de la siguiente manera:

```
SETS:
SUCURSAL /AREQUIPA,CUZCO,TRUJILLO/ : MINIMO, MAXIMO ;
PROYECTO /1..3/ ;
SUCURSAL_PROYECTO(SUCURSAL, PROYECTO): TASA, INVMAXIMA, ASIGNACION ;
ENDSETS
```


5. Definiendo la sección de DATA.

Presupuesto, es un parámetro aislado y no depende de ningún elemento, por lo tanto no es necesario definirlo en la sección SETS, directamente es definido en la sección data.

```
DATA:  
PRESUPUESTO = 35;  
ENDDATA
```

Los atributos MINIMO, MAXIMO, INVMAXIMA y TASA definidos anteriormente en la sección SETS, son atributos definidos para la lectura de parámetros, en cambio ASIGNACION es un atributo que se comporta como variable, por lo tanto este último no será definido en la sección DATA. Finalmente la sección DATA quedará de la siguiente manera:

```
DATA:  
  
PRESUPUESTO = 35;  
MINIMO = 8, 10, 15;  
MAXIMO = 0, 20, 0;  
  
INVMAXIMA = 8, 6, 7,  
            5, 8, 9,  
            10, 6, 15;  
  
TASA = 0.08, 0.06, 0.07,  
       0.05, 0.08, 0.09,  
       0.1, 0.06, 0.15;  
  
ENDDATA
```

Observación : Para la lectura del atributo MAXIMO, solo para la 2da sucursal existe límite máximo, para las otras que tienen cero, se debe asumir que no existe límite máximo, en la implementación de las restricciones se tratará este caso con filtros (*).

Note que cada una de las 3 sucursales tiene un valor asignado para un nivel de inversión MINIMO y MAXIMO, y cada una de las 9 combinaciones posibles de SUCURSAL y PROYECTO - es decir cada elemento de SUCURSAL_PROYECTO - tiene una inversión máxima INVMAXIMA y una TASA asociada. En la sección DATA, usted notará que no contiene valores para el atributo ASIGNACION. Esto es porque ASIGNACION representa las variables de decisión en el modelo, esto será determinado por LINGO.

6. Definiendo la estructura de las restricciones.

Definiendo la función Objetivo:

Formulación Matemática:

$$MAX = \sum_{i=1}^3 \sum_{j=1}^3 Tasa(i, j) * X_{(i,j)}$$

Formulación en LINGO:

```
MAX = @SUM( SUCURSAL_PROYECTO : TASA * ASIGNACION );
```

La función de **@SUM** es usada para alcanzar todos los miembros del SET derivado SUCURSAL_PROYECTO.

Ecuación Generada:

```
MAX .08 ASIGNACION( AREQUIPA, 1) + .06 ASIGNACION( AREQUIPA, 2)
+ .07 ASIGNACION( AREQUIPA, 3) + .05 ASIGNACION( CUZCO, 1)
+ .08 ASIGNACION( CUZCO, 2) + .09 ASIGNACION( CUZCO, 3)
+ .1 ASIGNACION( TRUJILLO, 1) + .06 ASIGNACION( TRUJILLO, 2)
+ .15 ASIGNACION( TRUJILLO, 3)
```

Restricción de uso mínimo de presupuesto por sucursal.

Formulación Matemática:

$$\sum_{j=1}^3 X_{(i,j)} \geq \text{Mínimo}_{(i)} \quad \forall i = 1,2,3$$

Formulación en LINGO:

```
@FOR( SUCURSAL(I) :
    @SUM( PROYECTO(J) :
        ASIGNACION(I,J) ) >= MINIMO(I) );
```

Para cada SUCURSAL I, la suma de las cantidades asignadas a cada SUCURSAL I asociadas al PROYECTO J, en la variable ASIGNACION(I,J), debe ser mayor o igual que el MINIMO de nivel de asignación de fondos de la SUCURSAL I.

Ecuación Generada:

```
ASIGNACION( AREQUIPA, 1) + ASIGNACION( AREQUIPA, 2)
+ ASIGNACION( AREQUIPA, 3) >= 8
```

```
ASIGNACION( CUZCO, 1) + ASIGNACION( CUZCO, 2)
+ ASIGNACION( CUZCO, 3) >= 10
```

```
ASIGNACION( TRUJILLO, 1) + ASIGNACION( TRUJILLO, 2)
+ ASIGNACION( TRUJILLO, 3) >= 15
```

Restricción de uso máximo de presupuesto por sucursal.

Formulación Matemática:

$$\sum_{j=1}^3 X_{(i,j)} \leq \text{Máximo}_{(i)} \quad \forall i = 2 \quad (\text{Solo para Cuzco})$$

Formulación en LINGO:

```
@FOR( SUCURSAL( I) | MAXIMO(I) #NE# 0 :  
  @SUM( PROYECTO (J):  
    ASIGNACION(I,J)) <= MAXIMO( I));
```

Solo se genera la ecuación para aquellos valores donde el atributo MAXIMO es distinto de cero, revise la observación anterior en definición de DATA (*)

Para cada SUCURSAL I, donde su valor de atributo nivel MAXIMO sea distinto de cero, la suma de las cantidades asignadas a cada SUCURSAL I asociadas al PROYECTO J, en la variable ASIGNACION(I,J), debe ser menor o igual que el MAXIMO de nivel de asignación de fondos de la SUCURSAL I.

Ecuación Generada:

Solo se genera la ecuación para la sucursal cuzco, que es la única que tiene una restricción de uso máximo de presupuesto de 20 millones. Para las otras sucursales que tienen cero (según la data leída en la sección DATA), se debe asumir que no existe restricción de límite máximo.

```
ASIGNACION( CUZCO, 1) + ASIGNACION( CUZCO, 2)  
  + ASIGNACION( CUZCO, 3) <= 20
```

Restricción de inversión máxima por proyecto en cada sucursal.

Formulación Matemática:

$$X_{(i,j)} \leq \text{InvMaxima}_{(i,j)} \quad \forall i = 1,2,3; \quad \forall j = 1,2,3$$

Formulación en LINGO:

```
@FOR( SUCURSAL_PROYECTO (I,J) :  
  ASIGNACION(I,J) <= INVMAXIMA(I,J) );
```

Para cada SUCURSAL I, y PROYECTO J, las cantidades asignadas a cada SUCURSAL I asociadas al PROYECTO J, en la variable ASIGNACION(I,J), debe ser menor o igual a la inversión máxima INVMAXIMA para la SUCURSAL I en el PROYECTO J.

Ecuación Generada:

```
ASIGNACION( AREQUIPA, 1) <= 8  
ASIGNACION( AREQUIPA, 2) <= 6  
ASIGNACION( AREQUIPA, 3) <= 7  
ASIGNACION( CUZCO, 1) <= 5  
ASIGNACION( CUZCO, 2) <= 8  
ASIGNACION( CUZCO, 3) <= 9  
ASIGNACION( TRUJILLO, 1) <= 10  
ASIGNACION( TRUJILLO, 2) <= 6  
ASIGNACION( TRUJILLO, 3) <= 15
```

Restricción de presupuesto disponible.

Formulación Matemática:

$$\sum_{i=1}^3 \sum_{j=1}^3 X_{(i,j)} \leq \text{PresupuestoTotal}$$

Formulación en LINGO:

```
@SUM( SUCURSAL_PROYECTO (I,J) :  
      ASIGNACION(I,J) ) <= PRESUPUESTO ;
```

La suma de las cantidades asignadas a cada SUCURSAL I asociadas al PROYECTO J, en la variable ASIGNACION(I,J), debe ser menor o igual que el presupuesto total PRESUPUESTOI.

Ecuación Generada:

```
ASIGNACION( AREQUIPA, 1) + ASIGNACION( AREQUIPA, 2)  
+ ASIGNACION( AREQUIPA, 3) + ASIGNACION( CUZCO, 1)  
+ ASIGNACION( CUZCO, 2) + ASIGNACION( CUZCO, 3)  
+ ASIGNACION( TRUJILLO, 1) + ASIGNACION( TRUJILLO, 2)  
+ ASIGNACION( TRUJILLO, 3) <= 35
```

7. Usando LINGO

- El proceso de resolver un programa matemático requiere un gran número de cálculos, los que son mejor ejecutados por los computadores.
- Al iniciar LINGO, este abre una ventana en blanco conocida como la ventana del modelo, en esta ventana pondremos “todo nuestro trabajo”.
- A través de los ejemplos desarrollados usaremos una serie de comandos y sentencias necesarias para trabajar sin problemas.
- El propósito principal de LINGO es permitirle al usuario:
 - Formular rápidamente un modelo.
 - Resolverlo.
 - Modificarlo

Desarrollando un Modelo

- En general un modelo de optimización consistirá de Función objetivo, Variables y Restricciones
- Los nombres de las variables deben comenzar con un carácter alfabético (A-Z), los caracteres subsecuentes pueden ser alfabéticos, numéricos o el carácter underline “_” hasta una longitud de 32 caracteres.
- Para Maximizar o Minimizar usamos Max o Min, seguido del signo =
- Cada línea en LINGO, terminará con un punto y coma “;”

- LINGO utiliza los caracteres \leq o \geq para denotar \leq y \geq
- Un comentario empieza con el carácter “!” y también debe finalizar con el carácter “;”
- LINGO no distingue entre mayúsculas y minúsculas

El modelo llevado al entorno LINGO

```

SETS:
! Conjuntos Primitivos;
SUCURSAL /AREQUIPA,CUZCO,TRUJILLO/: MINIMO, MAXIMO;
PROYECTO /1..3/: ;

! Conjuntos Derivados;
SUCURSAL_PROYECTO(SUCURSAL, PROYECTO): TASA, INVMAXIMA, ASIGNACION ;
ENDSETS

! Funcion Ojective;
MAX = @SUM( SUCURSAL_PROYECTO : TASA * ASIGNACION );

! Restriccion de uso minimo de presupuesto por sucursal;
@FOR( SUCURSAL(I):
    @SUM( PROYECTO(J) :
        ASIGNACION(I,J) ) >= MINIMO(I) );

! Restriccion de uso maximo de presupuesto por sucursal ;
@FOR( SUCURSAL( I) | MAXIMO(I) #NE# 0 :
    @SUM( PROYECTO (J):
        ASIGNACION(I,J) ) <= MAXIMO( I));

! Restriccion de inversion maxima por proyecto de sucursal ;
@FOR( SUCURSAL_PROYECTO (I,J) :
    ASIGNACION(I,J) <= INVMAXIMA(I,J) );

! Restriccion de presupuesto disponible;
@SUM( SUCURSAL_PROYECTO (I,J):
    ASIGNACION(I,J) ) <= PRESUPUESTO ;

DATA:
PRESUPUESTO = 35;
MINIMO = 8, 10 ,15;
MAXIMO = 0, 20, 0;

INVMAXIMA = 8, 6, 7,
            5, 8, 9,
            10, 6, 15;

TASA = 0.08, 0.06, 0.07,
        0.05, 0.08, 0.09,
        0.1, | 0.06, 0.15;

ENDDATA
END

```


Ahora resolvamos el modelo. En LINGO para Windows escoger solve del menú Lingo.

Resolviendo el Problema

Para resolver el problema haga un clic en el boton del bull ubicada en la barra de herramientas



La ventana "Solver Status"



State: Da el estado de la solución actual, los posibles estados, son:
"Global Optimum"
"Local Optimum"
"Feasible"
"Infeasible"
"Unbounded"
"Interrupted"
"Undetermined".

Iterations: Número de iteraciones realizadas.

Infeasibility: Cantidad por la cual las restricciones son violadas.

Objective: Valor actual de la función objetivo.

Best IP: Valor objetivo de la mejor solución entera encontrada.
(Solo modelos de programación entera).

IP Bound: Cota teórica de la función objetivo para modelos de Programación entera.

La solución aparece abajo:

Global optimal solution found at step: 13
Objective value: 3.980000

Variable	Value	Reduced Cost
PRESUPUESTO	35.00000	0.000000
MINIMO(AREQUIPA)	8.000000	0.000000
MINIMO(CUZCO)	10.00000	0.000000
MINIMO(TRUJILLO)	15.00000	0.000000
MAXIMO(AREQUIPA)	0.000000	0.000000
MAXIMO(CUZCO)	20.00000	0.000000
MAXIMO(TRUJILLO)	0.000000	0.000000
TASA(AREQUIPA, 1)	0.8000000E-01	0.000000
TASA(AREQUIPA, 2)	0.6000000E-01	0.000000
TASA(AREQUIPA, 3)	0.7000000E-01	0.000000
TASA(CUZCO, 1)	0.5000000E-01	0.000000
TASA(CUZCO, 2)	0.8000000E-01	0.000000
TASA(CUZCO, 3)	0.9000000E-01	0.000000
TASA(TRUJILLO, 1)	0.1000000	0.000000
TASA(TRUJILLO, 2)	0.6000000E-01	0.000000
TASA(TRUJILLO, 3)	0.1500000	0.000000
INVMAXIMA(AREQUIPA, 1)	8.000000	0.000000
INVMAXIMA(AREQUIPA, 2)	6.000000	0.000000
INVMAXIMA(AREQUIPA, 3)	7.000000	0.000000
INVMAXIMA(CUZCO, 1)	5.000000	0.000000
INVMAXIMA(CUZCO, 2)	8.000000	0.000000
INVMAXIMA(CUZCO, 3)	9.000000	0.000000
INVMAXIMA(TRUJILLO, 1)	10.00000	0.000000
INVMAXIMA(TRUJILLO, 2)	6.000000	0.000000
INVMAXIMA(TRUJILLO, 3)	15.00000	0.000000
ASIGNACION(AREQUIPA, 1)	8.000000	0.000000
ASIGNACION(AREQUIPA, 2)	0.000000	0.1000000E-01
ASIGNACION(AREQUIPA, 3)	0.000000	0.000000
ASIGNACION(CUZCO, 1)	0.000000	0.3000000E-01
ASIGNACION(CUZCO, 2)	1.000000	0.000000
ASIGNACION(CUZCO, 3)	9.000000	0.000000
ASIGNACION(TRUJILLO, 1)	2.000000	0.000000
ASIGNACION(TRUJILLO, 2)	0.000000	0.4000000E-01
ASIGNACION(TRUJILLO, 3)	15.00000	0.000000

Row	Slack or Surplus	Dual Price
1	3.980000	1.000000
2	0.000000	-0.3000000E-01
3	0.000000	-0.2000000E-01
4	2.000000	0.000000
5	10.00000	0.000000
6	0.000000	0.1000000E-01
7	6.000000	0.000000
8	7.000000	0.000000
9	5.000000	0.000000
10	7.000000	0.000000


11	0.0000000	0.1000000E-01
12	8.000000	0.0000000
13	6.000000	0.0000000
14	0.0000000	0.5000000E-01
15	0.0000000	0.1000000

El valor objetivo, en la parte superior del reporte, nos dice que la máxima rentabilidad es 3.98 millones de dólares.

El resultado nos da la asignación total de los recursos financieros, en el atributo ASIGNACION. El valor de ASIGNACION(Arequipa, 1), por ejemplo es 8. Esto significa que la cantidad de dinero asignado a la sucursal Arequipa para el proyecto 1 deberá ser 8 millones de dólares. Para ver la solución de sólo este atributo en LINGO, escoja la opción “solution” del menú “Lingo” e ingrese el nombre del atributo “ASIGNACION” en el cuadro de dialogo. También puede seleccionar el siguiente icono:



Y aparecerá la siguiente ventana:



Active con un check la opción que le permite mostrar solo los valores distintos de cero, y haga Click en OK. De esta manera, Lingo mostrará sólo los valores de las variables ASIGNACION distintos de cero, como se muestra abajo.

Variable	Value	Reduced Cost
ASIGNACION(AREQUIPA, 1)	8.000000	0.0000000
ASIGNACION(CUZCO, 2)	1.000000	0.0000000
ASIGNACION(CUZCO, 3)	9.000000	0.0000000
ASIGNACION(TRUJILLO, 1)	2.000000	0.0000000
ASIGNACION(TRUJILLO, 3)	15.00000	0.0000000

El siguiente modelo es una ampliación del modelo anterior, ilustra el uso del SET derivado no denso, usando una lista de miembros explícita.

PROBLEMA : MODELO DE ASIGNACION DE PROYECTO – (MODELO B)

Una compañía dispone de 180 millones como presupuesto para los siguientes 3 periodos, que se usaran entre sus 3 sucursales. Se ha establecido un nivel máximo de asignación de fondos de 20 millones para cada sucursal por periodo. En caso que una sucursal no exceda el límite del periodo, puede usar esa diferencia para el siguiente periodo. Cada sucursal tiene la oportunidad de dirigir distintos proyectos con los fondos que recibe. Para cada proyecto se ha establecido una tasa de ganancia (como un porcentaje de la inversión). Por otra parte, algunos de los proyectos obligan una inversión mínima. Determine la política de asignación que genere el mejor beneficio.

Inversión Mínima (millones)			
Sucursal	Proyecto	Periodo	Cantidad mínima
Arequipa	1	1	5
Cuzco	1	3	4
Trujillo	3	1	3

Sucursal	Periodo 1			Periodo 2			Periodo 3		
	Proyecto	Tasa de Utilidad (%)	Inversión Máxima (m)	Proyecto	Tasa de Utilidad (%)	Inversión Máxima (m)	Proyecto	Tasa de Utilidad (%)	Inversión Máxima (m)
Arequipa	1	8	10	1	9	9	1	8	10
	2	6	7	2	6	5	2	7	6
	3	7	9	3	7	9	3	7	9
Cuzco	1	5	7	1	6	4	1	5	7
	2	8	10	2	18	10	2	8	8
	3	9	4	3	9	3	3	9	4
Trujillo	1	10	6	1	11	6	1	7	6
	2	6	12	2	6	10	2	6	11
	3	15	6	3	15	5	3	15	6

Solución:

Formulación Algebraica del modelo

Definición de variables:

$X_{(i,j,k)}$ = Cantidad asignada a la sucursal i para el proyecto j en el periodo k .

$Sobrante_{(i,k)}$ = Cantidad sobrante, dinero no asignado a la sucursal i en el periodo k .

Definición de parámetros:

$Máximo_{(i,k)}$ = Nivel máximo de asignación de fondos para la sucursal i en el periodo k .

$\text{Minimo}_{(i,j,k)}$ = Nivel mínimo de asignación de fondos en la sucursal i , para el proyecto j , en el periodo k .

$\text{InvMaxima}_{(i,j,k)}$ = Inversión máxima en la sucursal i para el proyecto j en el periodo k .

PresupuestoTotal = Presupuesto total disponible.

$\text{Tasa}_{(i,j,k)}$ = Tasa de rentabilidad de la sucursal i , para el proyecto j durante el periodo k .

Restricción de uso máximo de presupuesto por periodo por sucursal.

$$\sum_{j=1}^3 X_{(i,j,k)} + \text{Sobrante}_{(i,k)} = \text{Máximo}_{(i,k)}$$

$\forall k = 1$ (Solo para el periodo $k=1$)
 $\forall i = 1,2,3$ (Sucursal)

$$\sum_{j=1}^3 X_{(i,j,k)} - \text{Sobrante}_{(i,k-1)} + \text{Sobrante}_{(i,k)} = \text{Máximo}_{(i,k)}$$

$\forall k = 2,3$ (Periodo)
 $\forall i = 1,2,3$ (Sucursal)

Restricción de inversión mínima para algunos proyectos específicos.

$$X_{(i,j,k)} \geq \text{Minimo}_{(i,j,k)} \quad \text{Solo para } (i=1, j=1, k=1) \wedge (i=2, j=1, k=3) \wedge (i=3, j=3, k=1)$$

Restricción de inversión máxima por proyecto en cada sucursal.

$$X_{(i,j,k)} \leq \text{InvMaxima}_{(i,j,k)} \quad \forall i = 1,2,3; \quad \forall j = 1,2,3 \quad \forall k = 1,2,3$$

Restricción de presupuesto disponible.

$$\sum_{i=1}^3 \sum_{j=1}^3 \sum_{k=1}^3 X_{(i,j,k)} \leq \text{PresupuestoTotal}$$

Función Objetivo

$$\text{MAX} = \sum_{i=1}^3 \sum_{j=1}^3 \sum_{k=1}^3 \text{Tasa}_{(i,j,k)} * X_{(i,j,k)}$$

Pasos para identificar y relacionar los elementos que participan en la estructura de un modelo en LINGO

Como este ejemplo es una ampliación del modelo anterior iremos de frente a la matriz de relaciones.

8. Matriz de elementos que participan en el modelo y sus relaciones, Indicar cuales son parámetros y variables.

Clasific.	ELEMENTOS	Sucur. (i)	Proy (j)	Period (k)	Tasa	Inv. Máximo	Mín	Máx	Presup	X (i,j,k)	Sobrante (i,k)
Parámetro	Sucursal (i)										
Parámetro	Proyecto (j)										
Parámetro	Periodo (k)										
Parámetro	Tasa	X	X	X							
Parámetro	Inv. Máxima	X	X	X							
Parámetro	Mínimo	X	X	X							
Parámetro	Máximo	X		X							
Parámetro	Presupuesto										
Variable	X (i,j,k)	X	X	X							
Variable	Sobrante.(i,k)	X		X							

9. Relacionar los elementos (“depende de”).

De la matriz anterior podemos observar que:

- Sucursal(i) no dependen de ningún elemento superior, por lo tanto es un elemento básico.
- Proyecto(j) no dependen de ningún elemento superior, por lo tanto es un elemento básico.
- Periodo (k) no dependen de ningún elemento superior, por lo tanto es un elemento básico.
- Tasa depende de los elementos Sucursal(i) , Proyecto(j) y Periodo(k).
- Inv. Máxima depende de los elementos Sucursal(i), Proyecto(j) y Periodo(k).
- Mínimo depende de los elemento Sucursal(i), Proyecto(j) y Periodo(k).
- Máximo depende de los elemento Sucursal(i) y Periodo(k).
- Presupuesto es un parámetro aislado y no depende de ningún elemento.
- $X_{(i,j,k)}$ es una variable que depende de los elementos Sucursal(i), Proyecto(j) y Periodo(k).
- $Sobrante.(i,k)$ es una variable que depende de los elemento Sucursal(i), y Periodo(k).

10. Relacionar los elementos (“depende de”) y plantear una estructura de SET.

Sucursal(i), no depende de ningún elemento superior, por lo tanto es un elemento básico, que se define como un conjunto primitivo, para nuestro ejemplo lo llamaremos el set SUCURSAL definido por extensión (Arequipa, Cuzco, Trujillo).

SETS:

SUCURSAL /AREQUIPA,CUZCO,TRUJILLO/ ;
ENDSETS

Proyecto(j), no depende de ningún elemento superior, por lo tanto es un elemento básico, que se define como un conjunto primitivo, para nuestro ejemplo lo llamaremos el set PROYECTO, compuesto por tres proyectos, y lo definiremos por comprensión.

```
SETS:
    SUCURSAL /AREQUIPA, CUZCO, TRUJILLO/: ;
    PROYECTO /1...3/: ;
ENDSETS
```

Periodo (k), no depende de ningún elemento superior, por lo tanto es un elemento básico, que se define como un conjunto primitivo, para nuestro ejemplo lo llamaremos el set PERIODO, compuesto por tres periodo, y lo definiremos por comprensión.

```
SETS:
    SUCURSAL /AREQUIPA, CUZCO, TRUJILLO/: ;
    PROYECTO /1...3/: ;
    PERIODO /1...3/: ;
ENDSETS
```

Los elementos $Sobrante.(i,k)$ y Máximo, depende de los elementos Sucursal(i) y Periodo(k) (definidos anteriormente como conjuntos), por lo tanto es necesario crear una estructura de un set derivado que permita enlazar ambos conjuntos; para nuestro ejemplo lo llamaremos el set SUCURSAL_PERIODO, compuesto por los conjuntos SUCURSAL y PERIODO, y lo definiremos como un conjunto derivado denso.

Así mismo el elemento $Sobrante.(i,k)$ y Máximo, serán llamados atributos del conjunto derivado SUCURSAL_PERIODO, y los definiremos como SOBRANTE y MAXIMO.

```
SETS:
    SUCURSAL /AREQUIPA, CUZCO, TRUJILLO/: ;
    PROYECTO /1...3/: ;
    PERIODO /1...3/: ;
    SUCURSAL_PERIODO(SUCURSAL, PERIODO): SOBRANTE, MAXIMO;
ENDSETS
```

Los elementos Tasa, Inv. Máxima y $X_{(i,j,k)}$, depende de los elementos Sucursal(i) , Proyecto(j) y Periodo(k) (definidos anteriormente como conjuntos), por lo tanto es necesario crear una estructura de un set derivado que permita enlazar los tres conjuntos; para nuestro ejemplo lo llamaremos el set SUCURSAL_PROYECTO_PERIODO, compuesto por los conjuntos SUCURSAL, PROYECTO y PERIODO, y lo definiremos como un conjunto derivado denso.

Así mismo los elementos Tasa, Inv. Máxima y $X_{(i,j,k)}$, serán llamados atributos del conjunto derivado SUCURSAL_PROYECTO_PERIODO, y los definiremos como TASA, INVMAXIMA y ASIGNACION.

```
SETS:
    SUCURSAL /AREQUIPA, CUZCO, TRUJILLO/: ;
    PROYECTO /1...3/: ;
    PERIODO /1...3/: ;
    SUCURSAL_PERIODO(SUCURSAL, PERIODO): SOBRANTE, MAXIMO;
    SUCURSAL_PROYECTO_PERIODO(SUCURSAL, PROYECTO, PERIODO):
        TASA, INVMAXIMA, ASIGNACION ;
ENDSETS
```

El elemento Mínimo depende de los elemento Sucursal(i), Proyecto(j) y Periodo(k) (definidos anteriormente como conjuntos), por lo tanto es necesario crear una estructura de un set derivado que permita enlazar los tres conjuntos; para nuestro ejemplo lo llamaremos el set INVERSION_MINIMA, compuesto por los conjuntos SUCURSAL, PROYECTO y PERIODO, y lo definiremos como un **conjunto derivado NO denso usando una lista de miembros explicita**, debido a que no todos los miembros de los tres conjuntos enlazados forman parte de este conjunto. Así mismo el elemento Mínimo, será llamado atributo del conjunto derivado INVERSION_MINIMA, y los definiremos como MINIMO. Finalmente la sección SETS quedara de la siguiente manera:

```
SETS:
  SUCURSAL /AREQUIPA, CUZCO, TRUJILLO/: ;
  PROYECTO /1...3/: ;
  PERIODO /1...3/: ;
  SUCURSAL_PERIODO(SUCURSAL, PERIODO): SOBRENTE, MAXIMO;
  SUCURSAL_PROYECTO_PERIODO(SUCURSAL, PROYECTO, PERIODO):
      TASA, INVMAXIMA, ASIGNACION ;
  INVERSION_MINIMA (SUCURSAL, PROYECTO, PERIODO)
      /AREQUIPA 1 1, CUZCO 1 3, TRUJILLO 3 1 /: MINIMO;
ENDSETS
```

11. Definiendo la sección de DATA.

Presupuesto, es un parámetro aislado y no depende de ningún elemento, por lo tanto no es necesario definirlo en la sección SETS, directamente es definido en la sección data.

```
DATA:
  PRESUPUESTO = 180;
  ENDDATA
```

Los atributos MINIMO, MAXIMO, INVMAXIMA y TASA definidos anteriormente en la secciona SETS, son atributos definidos para la lectura de parámetros, en cambio ASIGNACION y SOBRENTE son atributos que se comportan como variables, por lo tanto este ultimo no será definido en la sección DATA. Finalmente la sección DATA quedara de la siguiente manera:

```
DATA:

TASA =    0.08, 0.09, 0.08, 0.06, 0.06, 0.07, 0.07, 0.07, 0.07,
          0.05, 0.06, 0.05, 0.08, 0.18, 0.08, 0.09, 0.09, 0.09,
          0.10, 0.11, 0.07, 0.06, 0.06, 0.06, 0.15, 0.15, 0.15;

INVMAXIMA =    10, 9, 8, 7, 5, 7, 9, 9, 7,
               7, 4, 7, 10, 10, 8, 4, 3, 4,
               6, 6, 6, 12, 10, 11, 6, 5, 6;

PRESUPUESTO = 35;
MINIMO = 5, 4, 3;

MAXIMO = 20, 20, 20,
         20, 20, 20,
         20, 20, 20;

ENDDATA
```

Observación : En la sección DATA, usted notará que no contiene valores para el atributo ASIGNACION y SOBRANTE. Esto se da, por que estos atributos representan las variables de decisión en el modelo, esto será determinado por LINGO.

12. Definiendo la estructura de las restricciones.

Definiendo la función Objetivo:

Formulación Matemática:

$$MAX = \sum_{i=1}^3 \sum_{j=1}^3 \sum_{k=1}^3 Tasa_{(i,j,k)} * X_{(i,j,k)}$$

Formulación en LINGO:

`MAX = @SUM(SUCURSAL_PROYECTO_PERIODO: TASA * ASIGNACION);`

La función de **@SUM** es usada para alcanzar todos los miembros del SET derivado SUCURSAL_PROYECTO_PERIODO.

Ecuación Generada:

```
MAX .08 ASIGNACION(AREQUIPA, 1, 1)+ .09 ASIGNACION(AREQUIPA, 1, 2)
+ .08 ASIGNACION(AREQUIPA, 1, 3)+ .06 ASIGNACION(AREQUIPA, 2, 1)
+ .06 ASIGNACION(AREQUIPA, 2, 2)+ .07 ASIGNACION(AREQUIPA, 2, 3)
+ .07 ASIGNACION(AREQUIPA, 3, 1)+ .07 ASIGNACION(AREQUIPA, 3, 2)
+ .07 ASIGNACION(AREQUIPA, 3, 3) + .05 ASIGNACION(CUZCO, 1, 1)
+ .06 ASIGNACION(CUZCO, 1, 2) + .05 ASIGNACION(CUZCO, 1, 3)
+ .08 ASIGNACION(CUZCO, 2, 1) + .18 ASIGNACION(CUZCO, 2, 2)
+ .08 ASIGNACION(CUZCO, 2, 3) + .09 ASIGNACION(CUZCO, 3, 1)
+ .09 ASIGNACION(CUZCO, 3, 2) + .09 ASIGNACION(CUZCO, 3, 3)
+ .1 ASIGNACION(TRUJILLO, 1, 1)+ .11 ASIGNACION(TRUJILLO, 1, 2)
+ .07 ASIGNACION(TRUJILLO, 1, 3)+ .06 ASIGNACION(TRUJILLO, 2, 1)
+ .06 ASIGNACION(TRUJILLO, 2, 2)+ .06 ASIGNACION(TRUJILLO, 2, 3)
+ .15 ASIGNACION(TRUJILLO, 3, 1)+ .15 ASIGNACION(TRUJILLO, 3, 2)
+ .15 ASIGNACION(TRUJILLO, 3, 3)
```

Restricción de uso máximo de presupuesto por periodo por sucursal.

Formulación Matemática:

$$\sum_{j=1}^3 X_{(i,j,k)} + Sobrante_{(i,k)} = \text{Máximo}_{(i,k)}$$

$$\forall k = 1 \quad (\text{Solo para el periodo } k=1)$$

$$\forall i = 1,2,3 \quad (\text{Sucursal})$$

$$\sum_{j=1}^3 X_{(i,j,k)} - Sobrante_{(i,k-1)} + Sobrante_{(i,k)} = \text{Máximo}_{(i,k)}$$

$$\forall k = 2,3 \quad (\text{Periodo})$$

$$\forall i = 1,2,3 \quad (\text{Sucursal})$$

Formulación en LINGO:

```
@FOR( PERIODO(K) | K #EQ# 1 :  
  @FOR( SUCURSAL(I):  
    @SUM( PROYECTO(J) :  
      ASIGNACION(I,J,K) ) + SOBRANTE(I,K) = MAXIMO(I,K) ));
```

Solo para el PERIODO (k=1), y para cada SUCURSAL I, la suma de las cantidades asignadas a cada PROYECTO J asociado a la SUCURSAL I, en la variable ASIGNACION(i,j,k), mas el posible SONBRANTE (i,k) debe ser igual al MAXIMO del nivel de asignación de fondos de la SUCURSAL I, en el PERIODO K.

Ecuación Generada:

$$\text{SOBRANTE}(\text{AREQUIPA}, 1) + \text{ASIGNACION}(\text{AREQUIPA}, 1, 1) + \text{ASIGNACION}(\text{AREQUIPA}, 2, 1) + \text{ASIGNACION}(\text{AREQUIPA}, 3, 1) = 20$$

$$\text{SOBRANTE}(\text{CUZCO}, 1) + \text{ASIGNACION}(\text{CUZCO}, 1, 1) + \text{ASIGNACION}(\text{CUZCO}, 2, 1) + \text{ASIGNACION}(\text{CUZCO}, 3, 1) = 20$$

$$\text{SOBRANTE}(\text{TRUJILLO}, 1) + \text{ASIGNACION}(\text{TRUJILLO}, 1, 1) + \text{ASIGNACION}(\text{TRUJILLO}, 2, 1) + \text{ASIGNACION}(\text{TRUJILLO}, 3, 1) = 20$$

Formulación en LINGO:

```
@FOR( PERIODO(K) | K #NE# 1 :  
  @FOR( SUCURSAL(I):  
    @SUM( PROYECTO(J) :  
      ASIGNACION(I,J,K) ) - SOBRANTE(I,K-1) + SOBRANTE(I,K) =  
      MAXIMO(I,K) ));
```

Para los PERIODOS mayores que 1 (k >1), y para cada SUCURSAL I, la suma de las cantidades asignadas a cada PROYECTO J asociado a la SUCURSAL I, en la variable ASIGNACION(i,j,k), menos el SOBRANTE(i,k-1) del periodo anterior, mas el posible SONBRANTE (i,k) del periodo actual debe ser igual al MAXIMO del nivel de asignación de fondos de la SUCURSAL I, en el PERIODO K.

Ecuación Generada:

$$- \text{SOBRANTE}(\text{AREQUIPA}, 1) + \text{SOBRANTE}(\text{AREQUIPA}, 2) + \text{ASIGNACION}(\text{AREQUIPA}, 1, 2) + \text{ASIGNACION}(\text{AREQUIPA}, 2, 2) + \text{ASIGNACION}(\text{AREQUIPA}, 3, 2) = 20$$

$$- \text{SOBRANTE}(\text{CUZCO}, 1) + \text{SOBRANTE}(\text{CUZCO}, 2) + \text{ASIGNACION}(\text{CUZCO}, 1, 2) + \text{ASIGNACION}(\text{CUZCO}, 2, 2) + \text{ASIGNACION}(\text{CUZCO}, 3, 2) = 20$$

$$- \text{SOBRANTE}(\text{TRUJILLO}, 1) + \text{SOBRANTE}(\text{TRUJILLO}, 2) + \text{ASIGNACION}(\text{TRUJILLO}, 1, 2) + \text{ASIGNACION}(\text{TRUJILLO}, 2, 2) + \text{ASIGNACION}(\text{TRUJILLO}, 3, 2) = 20$$

$$- \text{SOBRANTE}(\text{AREQUIPA}, 2) + \text{SOBRANTE}(\text{AREQUIPA}, 3) + \text{ASIGNACION}(\text{AREQUIPA}, 1, 3) + \text{ASIGNACION}(\text{AREQUIPA}, 2, 3) + \text{ASIGNACION}(\text{AREQUIPA}, 3, 3) = 20$$

$$- \text{SOBRANTE}(\text{CUZCO}, 2) + \text{SOBRANTE}(\text{CUZCO}, 3) + \text{ASIGNACION}(\text{CUZCO}, 1, 3) + \text{ASIGNACION}(\text{CUZCO}, 2, 3) + \text{ASIGNACION}(\text{CUZCO}, 3, 3) = 20$$

- SOBRENTE(TRUJILLO, 2) + SOBRENTE(TRUJILLO, 3) +
 ASIGNACION(TRUJILLO, 1, 3) + ASIGNACION(TRUJILLO, 2, 3) +
 ASIGNACION(TRUJILLO, 3, 3) = 20

Restricción de inversión mínima para algunos proyectos específicos.

Formulación Matemática:

$$X_{(i,j,k)} \geq \text{Minimo}_{(i,j,k)} \quad \text{Solo para } (i=1, j=1, k=1) \wedge (i=2, j=1, k=3) \wedge (i=3, j=3, k=1)$$

Formulación en LINGO:

```
@FOR( INVERSION_MINIMA( I, J, K) :
  ASIGNACION( I, J, K) >= MINIMO( I, J, K) );
```

Para cada SUCURSAL i, PROYECTO j y PERIODO k, del conjunto derivado INVERSION_MINIMA, - recuerde que se definió como un conjunto derivado no denso usando una lista de miembros explicita - las cantidades asignadas a cada SUCURSAL i asociadas al PROYECTO j y al PERIODO k, en la variable ASIGNACION(i,j), debe ser mayor o igual a la inversión MINIMA en la SUCURSAL i, para el PROYECTO J, en el PERIODO k.

Recuerde que la definición del conjunto INVERSION_MINIMA solo tiene 3 elementos y su definición es:

```
INVERSION_MINIMA( SUCURSAL, PROYECTO, PERIODO)
  /AREQUIPA 1 1, CUZCO 1 3, TRUJILLO 3 1 /: MINIMO;
```

Ecuación Generada:

ASIGNACION(AREQUIPA, 1, 1) >= 5
 ASIGNACION(CUZCO, 1, 3) >= 4
 ASIGNACION(TRUJILLO, 3, 1) >= 3

Restricción de inversión máxima por proyecto en cada sucursal.

Formulación Matemática:

$$X_{(i,j,k)} \leq \text{InvMaxima}_{(i,j,k)} \quad \forall i = 1,2,3; \quad \forall j = 1,2,3 \quad \forall k = 1,2,3$$

Formulación en LINGO:

```
@FOR( SUCURSAL_PROYECTO_PERIODO( I, J, K) :
  ASIGNACION( I, J, K) <= INVMAXIMA( I, J, K) );
```


Ecuación Generada:

```
ASIGNACION( AREQUIPA, 1, 1) <= 10
ASIGNACION( AREQUIPA, 1, 2) <= 9
ASIGNACION( AREQUIPA, 1, 3) <= 8
ASIGNACION( AREQUIPA, 2, 1) <= 7
ASIGNACION( AREQUIPA, 2, 2) <= 5
ASIGNACION( AREQUIPA, 2, 3) <= 7
ASIGNACION( AREQUIPA, 3, 1) <= 9
ASIGNACION( AREQUIPA, 3, 2) <= 9
ASIGNACION( AREQUIPA, 3, 3) <= 7
ASIGNACION( CUZCO, 1, 1) <= 7
ASIGNACION( CUZCO, 1, 2) <= 4
ASIGNACION( CUZCO, 1, 3) <= 7
ASIGNACION( CUZCO, 2, 1) <= 10
ASIGNACION( CUZCO, 2, 2) <= 10
ASIGNACION( CUZCO, 2, 3) <= 8
ASIGNACION( CUZCO, 3, 1) <= 4
ASIGNACION( CUZCO, 3, 2) <= 3
ASIGNACION( CUZCO, 3, 3) <= 4
ASIGNACION( TRUJILLO, 1, 1) <= 6
ASIGNACION( TRUJILLO, 1, 2) <= 6
ASIGNACION( TRUJILLO, 1, 3) <= 6
ASIGNACION( TRUJILLO, 2, 1) <= 12
ASIGNACION( TRUJILLO, 2, 2) <= 10
ASIGNACION( TRUJILLO, 2, 3) <= 11
ASIGNACION( TRUJILLO, 3, 1) <= 6
ASIGNACION( TRUJILLO, 3, 2) <= 5
ASIGNACION( TRUJILLO, 3, 3) <= 6
```

Restricción de presupuesto disponible.

Formulación Matemática:

$$\sum_{i=1}^3 \sum_{j=1}^3 \sum_{k=1}^3 X_{(i,j,k)} \leq \text{PresupuestoTotal}$$

Formulación en LINGO:

```
@SUM( SUCURSAL_PROYECTO_PERIODO(I,J,K):
      ASIGNACION(I,J,K)) <= PRESUPUESTO ;
```

Ecuación Generada:

```
ASIGNACION( AREQUIPA, 1, 1) + ASIGNACION( AREQUIPA, 1, 2)
+ ASIGNACION( AREQUIPA, 1, 3) + ASIGNACION( AREQUIPA, 2, 1)
+ ASIGNACION( AREQUIPA, 2, 2) + ASIGNACION( AREQUIPA, 2, 3)
+ ASIGNACION( AREQUIPA, 3, 1) + ASIGNACION( AREQUIPA, 3, 2)
+ ASIGNACION( AREQUIPA, 3, 3) + ASIGNACION( CUZCO, 1, 1)
+ ASIGNACION( CUZCO, 1, 2) + ASIGNACION( CUZCO, 1, 3)
+ ASIGNACION( CUZCO, 2, 1) + ASIGNACION( CUZCO, 2, 2)
+ ASIGNACION( CUZCO, 2, 3) + ASIGNACION( CUZCO, 3, 1)
+ ASIGNACION( CUZCO, 3, 2) + ASIGNACION( CUZCO, 3, 3)
+ ASIGNACION( TRUJILLO, 1, 1) + ASIGNACION( TRUJILLO, 1, 2)
+ ASIGNACION( TRUJILLO, 1, 3) + ASIGNACION( TRUJILLO, 2, 1)
+ ASIGNACION( TRUJILLO, 2, 2) + ASIGNACION( TRUJILLO, 2, 3)
+ ASIGNACION( TRUJILLO, 3, 1) + ASIGNACION( TRUJILLO, 3, 2)
+ ASIGNACION( TRUJILLO, 3, 3) <= 180
```

El modelo llevado al entorno LINGO

```
MODEL:
! MODELO DE ASIGNACION DE PRESUPUESTO;
SETS:

! Conjuntos Primitivos;
SUCURSAL /AREQUIPA,CUZCO,TRUJILLO/;;
PROYECTO /1..3/: ;
PERIODO /1..3/: ;

! Conjuntos Derivados;
SUCURSAL_PERIODO (SUCURSAL,PERIODO) : SOBRANTE, MAXIMO;
SUCURSAL_PROYECTO_PERIODO (SUCURSAL,PROYECTO,PERIODO):TASA,
INVMAXIMA, ASIGNACION;
INVERSION_MINIMA (SUCURSAL,PROYECTO,PERIODO)
/AREQUIPA 1 1, CUZCO 1 3, TRUJILLO 3 1 /: MINIMO;
ENDSETS

! Funcion Objetivo;
MAX = @SUM(SUCURSAL_PROYECTO_PERIODO: TASA * ASIGNACION );

! Restriccion de uso maximo de presupuesto por periodo por sucursal ;

@FOR(PERIODO(K) | K #EQ# 1 :
@FOR( SUCURSAL(I):
@SUM( PROYECTO(J) :
ASIGNACION(I,J,K) ) + SOBRANTE(I,K) = MAXIMO(I,K) ));

@FOR(PERIODO(K) | K #NE# 1 :
@FOR( SUCURSAL(I):
@SUM( PROYECTO(J) :
ASIGNACION(I,J,K) ) - SOBRANTE(I,K-1)+ SOBRANTE(I,K) =
MAXIMO(I,K) ));

! Restriccion de inversion minima para algunos proyectos especificos ;

@FOR(INVERSION_MINIMA(I,J,K):
ASIGNACION(I,J,K) >= MINIMO(I,J,K));

! Restriccion de inversion maxima por proyecto en cada sucursal ;
@FOR(SUCURSAL_PROYECTO_PERIODO(I,J,K):
ASIGNACION(I,J,K) <= INVMAXIMA(I,J,K) );

! Restriccion de presupuesto disponible;
@SUM( SUCURSAL_PROYECTO_PERIODO(I,J,K):
ASIGNACION(I,J,K)) <= PRESUPUESTO ;

DATA:
TASA = 0.08, 0.09, 0.08, 0.06, 0.06, 0.07, 0.07, 0.07, 0.07,
0.05, 0.06, 0.05, 0.08, 0.18, 0.08, 0.09, 0.09, 0.09,
0.10, 0.11, 0.07, 0.06, 0.06, 0.06, 0.15, 0.15, 0.15;

INVMAXIMA = 10, 9, 8, 7, 5, 7, 9, 9, 7,
7, 4, 7, 10, 10, 8, 4, 3, 4,
6, 6, 6, 12, 10, 11, 6, 5, 6;

MINIMO = 5, 4, 3;

PRESUPUESTO = 180;
```

```

MAXIMO = 20, 20, 20,
         20, 20, 20,
         20, 20, 20;

```

ENDDATA

END

Ahora resolvamos el modelo. En LINGO seleccionar la opción solve del menú Lingo.

Resolviendo el Problema

La solución aparece abajo:

```

Global optimal solution found at step:          43
Objective value:                               15.40000

```

El valor objetivo, en la parte superior del reporte, nos dice que la máxima rentabilidad es 15.4 millones de dólares.

Para ver la solución de sólo los atributos variables, seleccione la opción “solution” del menú “Lingo” e ingrese el nombre del atributo “ASIGNACION”, “SOBRANTE” en el cuadro de dialogo.

Variable	Value	Reduced Cost
ASIGNACION(AREQUIPA, 1, 1)	10.00000	0.000000
ASIGNACION(AREQUIPA, 1, 2)	9.000000	0.000000
ASIGNACION(AREQUIPA, 1, 3)	8.000000	0.000000
ASIGNACION(AREQUIPA, 2, 2)	1.000000	0.000000
ASIGNACION(AREQUIPA, 2, 3)	7.000000	0.000000
ASIGNACION(AREQUIPA, 3, 1)	9.000000	0.000000
ASIGNACION(AREQUIPA, 3, 2)	9.000000	0.000000
ASIGNACION(AREQUIPA, 3, 3)	7.000000	0.000000
ASIGNACION(CUZCO, 1, 1)	6.000000	0.000000
ASIGNACION(CUZCO, 1, 2)	4.000000	0.000000
ASIGNACION(CUZCO, 1, 3)	7.000000	0.000000
ASIGNACION(CUZCO, 2, 1)	10.00000	0.000000
ASIGNACION(CUZCO, 2, 2)	10.00000	0.000000
ASIGNACION(CUZCO, 2, 3)	8.000000	0.000000
ASIGNACION(CUZCO, 3, 1)	4.000000	0.000000
ASIGNACION(CUZCO, 3, 2)	3.000000	0.000000
ASIGNACION(CUZCO, 3, 3)	4.000000	0.000000
ASIGNACION(TRUJILLO, 1, 1)	6.000000	0.000000
ASIGNACION(TRUJILLO, 1, 2)	6.000000	0.000000
ASIGNACION(TRUJILLO, 1, 3)	6.000000	0.000000
ASIGNACION(TRUJILLO, 2, 1)	7.000000	0.000000
ASIGNACION(TRUJILLO, 2, 2)	10.00000	0.000000
ASIGNACION(TRUJILLO, 2, 3)	8.000000	0.000000
ASIGNACION(TRUJILLO, 3, 1)	6.000000	0.000000
ASIGNACION(TRUJILLO, 3, 2)	5.000000	0.000000
ASIGNACION(TRUJILLO, 3, 3)	6.000000	0.000000

Variable	Value	Reduced Cost
SOBRANTE(AREQUIPA, 1)	1.000000	0.000000
SOBRANTE(AREQUIPA, 2)	2.000000	0.000000
SOBRANTE(CUZCO, 2)	3.000000	0.000000
SOBRANTE(CUZCO, 3)	4.000000	0.000000
SOBRANTE(TRUJILLO, 1)	1.000000	0.000000

PROBLEMA : MODELO DE ASIGNACION DE PROYECTO – (MODELO C)

Una compañía dispone de 180 millones como presupuesto para los siguientes 3 periodos, que se usaran entre sus 3 sucursales. Se ha establecido un nivel máximo de asignación de fondos de 20 millones para cada sucursal por periodo. En caso que una sucursal no exceda el límite del periodo, puede usar esa diferencia para el siguiente periodo. Cada sucursal tiene la oportunidad de dirigir distintos proyectos con los fondos que recibe. Para cada proyecto se ha establecido una tasa de ganancia (como un porcentaje de la inversión). Por otra parte, algunos de los proyectos permiten una inversión mínima. Determine la política de asignación que genere el mejor beneficio.

Inversión Mínima (millones)			
Sucursal	Proyecto	Periodo	Cantidad mínima
Arequipa	1	1	5
Cuzco	1	3	4
Trujillo	3	1	3

Sucursal	Periodo 1			Periodo 2			Periodo 3		
	Proyecto	Tasa de Utilidad (%)	Inversión Máxima (m)	Proyecto	Tasa de Utilidad (%)	Inversión Máxima (m)	Proyecto	Tasa de Utilidad (%)	Inversión Máxima (m)
Arequipa	1	8	10	1	9	9	1	8	10
	2	6	7	2	6	5	2	7	6
	3	7	9	3	7	9	3	7	9
Cuzco	1	5	7	1	6	4	1	5	7
	2	8	10	2	18	10	2	8	8
	3	9	4	3	9	3	3	9	4
Trujillo	1	10	6	1	11	6	1	7	6
	2	6	12	2	6	10	2	6	11
	3	15	6	3	15	5	3	15	6

Adicionalmente por una directiva de la alta gerencia se requiere que los proyectos que tienen una rentabilidad mayotes del 5% por lo menos tengan una inversión de 90 millones.

Solución:

Formulación Algebraica del modelo

Definición de variables:

$X_{(i,j,k)}$ = Cantidad asignada a la sucursal i para el proyecto j en el periodo k .

$Sobrante_{(i,k)}$ = Cantidad sobrante, dinero no asignado a la sucursal i en el periodo k .

Definición de parámetros:

$Máximo_{(i,k)}$ = Nivel máximo de asignación de fondos para la sucursal i en el periodo k .

$Mínimo_{(i,j,k)}$ = Nivel mínimo de asignación de fondos en la sucursal i , para el proyecto j , en el periodo k .

$InvMaxima_{(i,j,k)}$ = Inversión máxima en la sucursal i para el proyecto j en el periodo k .

PresupuestoTotal = Presupuesto total disponible.

Tasa_(i,j,k) = Tasa de rentabilidad de la sucursal i, para el proyecto j durante el periodo k.

Restricción de uso máximo de presupuesto por periodo por sucursal.

$$\sum_{j=1}^3 X_{(i,j,k)} + Sobrante_{(i,k)} = \text{Máximo}_{(i,k)}$$

$\forall k = 1$ (Solo para el periodo k=1)
 $\forall i = 1,2,3$ (Sucursal)

$$\sum_{j=1}^3 X_{(i,j,k)} - Sobrante_{(i,k-1)} + Sobrante_{(i,k)} = \text{Máximo}_{(i,k)}$$

$\forall k = 2,3$ (Periodo)
 $\forall i = 1,2,3$ (Sucursal)

Restricción de inversión mínima para algunos proyectos específicos.

$$X_{(i,j,k)} \geq \text{Minimo}_{(i,j,k)} \quad \text{Solo para } (i=1, j=1, k=1) \wedge (i=2, j=1, k=3) \wedge (i=3, j=3, k=1)$$

Restricción de inversión máxima por proyecto en cada sucursal.

$$X_{(i,j,k)} \leq \text{InvMaxima}_{(i,j,k)} \quad \forall i = 1,2,3; \quad \forall j = 1,2,3 \quad \forall k = 1,2,3$$

Restricción de presupuesto disponible.

$$\sum_{i=1}^3 \sum_{j=1}^3 \sum_{k=1}^3 X_{(i,j,k)} \leq \text{PresupuestoTotal}$$

Función Objetivo

$$\text{MAX} = \sum_{i=1}^3 \sum_{j=1}^3 \sum_{k=1}^3 \text{Tasa}_{(i,j,k)} * X_{(i,j,k)}$$

Pasos para identificar y relacionar los elementos que participan en la estructura de un modelo en LINGO

Como este ejemplo es una ampliación del modelo anterior iremos directamente a la matriz de relaciones.

13. Matriz de elementos que participan en el modelo y sus relaciones, Indicar cuales son parámetros y variables.

Clasific.	ELEMENTOS	Sucur. (i)	Proy (j)	Period (k)	Tasa	Inv. Máximo	Mín	Máx	Presup	X (i,j,k)	Sobrante (i,k)
Parámetro	Sucursal (i)										
Parámetro	Proyecto (j)										
Parámetro	Periodo (k)										
Parámetro	Tasa	X	X	X							
Parámetro	Inv. Máxima	X	X	X							
Parámetro	Mínimo	X	X	X							
Parámetro	Máximo	X		X							
Parámetro	Presupuesto										
Variable	X (i,j,k)	X	X	X							
Variable	Sobrante.(i,k)	X		X							

14. Relacionar los elementos (“depende de”).

De la matriz anterior podemos observar que:

- Sucursal(i) no dependen de ningún elemento superior, por lo tanto es un elemento básico.
- Proyecto(j) no dependen de ningún elemento superior, por lo tanto es un elemento básico.
- Periodo (k) no dependen de ningún elemento superior, por lo tanto es un elemento básico.
- Tasa depende de los elementos Sucursal(i) , Proyecto(j) y Periodo(k).
- Inv. Máxima depende de los elementos Sucursal(i), Proyecto(j) y Periodo(k).
- Mínimo depende de los elemento Sucursal(i), Proyecto(j) y Periodo(k).
- Máximo depende de los elemento Sucursal(i) y Periodo(k).
- Presupuesto es un parámetro aislado y no depende de ningún elemento.
- $X_{(i,j,k)}$ es una variable que depende de los elementos Sucursal(i), Proyecto(j) y Periodo(k).
- $Sobrante.(i,k)$ es una variable que depende de los elemento Sucursal(i), y Periodo(k).

15. Relacionar los elementos (“depende de”) y plantear una estructura de SET.

Sucursal(i), no depende de ningún elemento superior, por lo tanto es un elemento básico, que se define como un conjunto primitivo, para nuestro ejemplo lo llamaremos el set SUCURSAL definido por extensión (Arequipa, Cuzco, Trujillo).

SETS:

SUCURSAL /AREQUIPA,CUZCO,TRUJILLO/ ;
ENDSETS

Proyecto(j), no depende de ningún elemento superior, por lo tanto es un elemento básico, que se define como un conjunto primitivo, para nuestro ejemplo lo llamaremos el set PROYECTO, compuesto por tres proyectos, y lo definiremos por comprensión.

```
SETS:
    SUCURSAL /AREQUIPA, CUZCO, TRUJILLO/: ;
    PROYECTO /1...3/: ;
ENDSETS
```

Periodo (k), no depende de ningún elemento superior, por lo tanto es un elemento básico, que se define como un conjunto primitivo, para nuestro ejemplo lo llamaremos el set PERIODO, compuesto por tres periodo, y lo definiremos por comprensión.

```
SETS:
    SUCURSAL /AREQUIPA, CUZCO, TRUJILLO/: ;
    PROYECTO /1...3/: ;
    PERIODO /1...3/: ;
ENDSETS
```

Los elementos $Sobrante.(i,k)$ y Máximo, depende de los elementos Sucursal(i) y Periodo(k) (definidos anteriormente como conjuntos), por lo tanto es necesario crear una estructura de un set derivado que permita enlazar ambos conjuntos; para nuestro ejemplo lo llamaremos el set SUCURSAL_PERIODO, compuesto por los conjuntos SUCURSAL y PERIODO, y lo definiremos como un conjunto derivado denso.

Así mismo el elemento $Sobrante.(i,k)$ y Máximo, serán llamados atributos del conjunto derivado SUCURSAL_PERIODO, y los definiremos como SOBRANTE y MAXIMO.

```
SETS:
    SUCURSAL /AREQUIPA, CUZCO, TRUJILLO/: ;
    PROYECTO /1...3/: ;
    PERIODO /1...3/: ;
    SUCURSAL_PERIODO(SUCURSAL, PERIODO): SOBRANTE, MAXIMO;
ENDSETS
```

Los elementos Tasa, Inv. Máxima y $X_{(i,j,k)}$, depende de los elementos Sucursal(i) , Proyecto(j) y Periodo(k) (definidos anteriormente como conjuntos), por lo tanto es necesario crear una estructura de un set derivado que permita enlazar los tres conjuntos; para nuestro ejemplo lo llamaremos el set SUCURSAL_PROYECTO_PERIODO, compuesto por los conjuntos SUCURSAL, PROYECTO y PERIODO, y lo definiremos como un conjunto derivado denso.

Así mismo los elementos Tasa, Inv. Máxima y $X_{(i,j,k)}$, serán llamados atributos del conjunto derivado SUCURSAL_PROYECTO_PERIODO, y los definiremos como TASA, INVMAXIMA y ASIGNACION.

```

SETS:
  SUCURSAL /AREQUIPA, CUZCO, TRUJILLO/: ;
  PROYECTO /1...3/: ;
  PERIODO /1...3/: ;
  SUCURSAL_PERIODO(SUCURSAL, PERIODO): SOBRANTE, MAXIMO;
  SUCURSAL_PROYECTO_PERIODO(SUCURSAL, PROYECTO, PERIODO):
    TASA, INVMAXIMA, ASIGNACION ;
ENDSETS

```

El elemento Mínimo depende de los elemento Sucursal(i), Proyecto(j) y Periodo(k) (definidos anteriormente como conjuntos), por lo tanto es necesario crear una estructura de un set derivado que permita enlazar los tres conjuntos; para nuestro ejemplo lo llamaremos el set INVERSION_MINIMA, compuesto por los conjuntos SUCURSAL, PROYECTO y PERIODO, y lo definiremos como un **conjunto derivado NO denso usando una lista de miembros explícita**, debido a que no todos los miembros de los tres conjuntos enlazados forman parte de este conjunto. Así mismo el elemento Mínimo, será llamado atributo del conjunto derivado INVERSION_MINIMA, y los definiremos como MINIMO. Finalmente la sección SETS quedara de la siguiente manera:

```

SETS:
  SUCURSAL /AREQUIPA, CUZCO, TRUJILLO/: ;
  PROYECTO /1...3/: ;
  PERIODO /1...3/: ;
  SUCURSAL_PERIODO(SUCURSAL, PERIODO): SOBRANTE, MAXIMO;
  SUCURSAL_PROYECTO_PERIODO(SUCURSAL, PROYECTO, PERIODO):
    TASA, INVMAXIMA, ASIGNACION ;
  INVERSION_MINIMA (SUCURSAL, PROYECTO, PERIODO)
    /AREQUIPA 1 1, CUZCO 1 3, TRUJILLO 3 1 /: MINIMO;
ENDSETS

```

16. Definiendo la sección de DATA.

Presupuesto, es un parámetro aislado y no depende de ningún elemento, por lo tanto no es necesario definirlo en la sección SETS, directamente es definido en la sección data.

```

DATA:
  PRESUPUESTO = 180;
ENDDATA

```

Los atributos MINIMO, MAXIMO, INVMAXIMA y TASA definidos anteriormente en la secciona SETS, son atributos definidos para la lectura de parámetros, en cambio ASIGNACION y SOBRANTE son atributos que se comportan como variables, por lo tanto este ultimo no será definido en la sección DATA. Finalmente la sección DATA quedara de la siguiente manera:

```

DATA:
  TASA = 0.08, 0.09, 0.08, 0.06, 0.06, 0.07, 0.07, 0.07, 0.07,
        0.05, 0.06, 0.05, 0.08, 0.18, 0.08, 0.09, 0.09, 0.09,
        0.10, 0.11, 0.07, 0.06, 0.06, 0.06, 0.15, 0.15, 0.15;

  INVMAXIMA = 10, 9, 8, 7, 5, 7, 9, 9, 7,
             7, 4, 7, 10, 10, 8, 4, 3, 4,
             6, 6, 6, 12, 10, 11, 6, 5, 6;

```


PRESUPUESTO = 35;
MINIMO = 5, 4, 3;

MAXIMO = 20, 20, 20,
 20, 20, 20,
 20, 20, 20;

ENDDATA

Observaciones: En la sección DATA, usted notará que no contiene valores para el atributo ASIGNACION y SOBRENTE. Esto se da, por que estos atributos representan las variables de decisión en el modelo, esto será determinado por LINGO.